

AD-A066 256

MINIMUF-35: IMPROVED VERSION OF MINIMUF-3 A SIMPLIFIED
HF MUF PREDICTION ALGORITHM(U) NAVAL OCEAN SYSTEMS
CENTER SAN DIEGO CA R B ROSE ET AL. 26 OCT 78

1/1

UNCLASSIFIED

NOSC-TD-201 GIDEP-E150-1309

F/G 20/14

NL



FILMED
IN
DTIC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS - 1963 - A

NOSC

ADDITIONAL COPY

AD-A066256

NOSC TD 201

NOSC TD 201

Technical Document 201

MINIMUF-3.5 Improved version of MINIMUF-3, a simplified hf MUF prediction algorithm

RB Rose
JN Martin

26 October 1978

Prepared for
Naval Security Group (G53)

Approved for public release; distribution unlimited

**NAVAL OCEAN SYSTEMS CENTER
SAN DIEGO, CALIFORNIA 92152**



NAVAL OCEAN SYSTEMS CENTER, SAN DIEGO, CA 92152

AN ACTIVITY OF THE NAVAL MATERIAL COMMAND

RR GAVAZZI, CAPT, USN

Commander

HL BLOOD

Technical Director

ADMINISTRATIVE INFORMATION

Work was performed with funding provided by the Naval Security Group (G53) under the CLASSIC PROPHET project (PE OMN, Project NSG, task area 0, NOSC work unit MP19), by members of the Special Projects Group.

Released by
JH Richter, Head
EM Propagation Division

Under authority of
JD Hightower, Head
Environmental Sciences Department

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER NOSC Technical Document 201 (TD 201)	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) MINIMUF-3.5 Improved version of MINIMUF-3, a simplified hf MUF prediction algorithm		5. TYPE OF REPORT & PERIOD COVERED
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) RB Rose JN Martin		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Naval Ocean Systems Center San Diego, CA 92152		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS OMN, NSG, 0, MP19
11. CONTROLLING OFFICE NAME AND ADDRESS Naval Security Group (G53) Washington, DC 20390		12. REPORT DATE 26 October 1978
		13. NUMBER OF PAGES
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE
16. DISTRIBUTION STATEMENT (of this Report) Approved for public release; distribution unlimited		
17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)		
18. SUPPLEMENTARY NOTES		
19. KEY WORDS (Continue on reverse side if necessary and identify by block number) MINIMUF Minimum usable frequency MUF prediction		
20. ABSTRACT (Continue on reverse side if necessary and identify by block number) Within an 800-8000 km range, the original MINIMUF-3 reported in NOSC TR 186 will predict hf MUF within specified confidence limits. But the development of new applications involving very long transmission paths uncovered several bugs. Corrections are reported here that do not affect the original accuracy limits but do allow MINIMUF to be used out to the antipodal point. This document presents the most current model, called MINIMUF-3.5, describes the improvements, and presents a data set with which each user can check out his version of MINIMUF-3.5.		


DD FORM 1473
1 JAN 73EDITION OF 1 NOV 65 IS OBSOLETE
S/N 0102-LF-014-6601

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)



UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE(When Data Entered)

INTRODUCTION

The initial report on the MINIMUF-3 model (ref 1) found widespread interest throughout the hf community. That model has been successfully adapted for use on a variety of minicomputers and microcomputers and has been used in numerous ways. As long as the 800-8000 km range constraint discussed in ref 1 is not exceeded, the original MINIMUF-3 will perform within confidence limits specified in the first verification study. But the development of new applications involving very long transmission paths uncovered several bugs that had to be corrected. The changes made do not affect the original accuracy limits but do allow MINIMUF to be used out to the antipodal point. The purposes of this document are to present the most current model, called MINIMUF-3.5, to describe the improvements made, and to present a data set with which each user can check out his version of MINIMUF-3.5. The original report, ref 1, explains the theory behind this model.

CHANGES TO ORIGINAL MINIMUF MODEL

The model listed in the BASIC program on page 41 of reference 1 contained two problems, as follows.

1. Line number 891 is a conditional branch:

```
891 IF COS(L0+Y2)>-0.26 THEN 900
```

If the branch to line 900 is not taken, subsequent execution of the line

```
1220 G2=G2*(1-0.1*EXP((K9-24)/3))
```

references the variable K9, which will not have been properly defined.

The problem is eliminated by insertion of the following line of code:

```
892 K9=0
```

2. For very long path lengths, the computed MUF is unrealistically low. For paths sufficiently long, a fatal error can occur at line 898 or line 1000 with a negative argument for the square root function. This is because the value of M9 (computed in line 897 or 990) reaches a maximum value for a path length of about 8000 km (the longest path used in the verification tests) and then decreases in magnitude, becoming negative for path lengths of about 16 000 km. The following changes correct this problem:

Delete lines 897, 898, 990 and 1000. Insert the lines

```
897 M9=SIN(2.5*G1*K5 MIN PI/2)
```

```
898 M9=1+2.5*M9*SQR(M9)
```

```
990 M9=SIN(2.5*G1*K5 MIN PI/2)
```

```
1000 M9=1+2.5*M9*SQR(M9)
```

In addition to these changes, MINIMUF-3.5 uses a somewhat different solution for the spherical trigonometry problems involved in computing path length and control point coordinates. This change simplifies the code and eliminates the need for GOSUBs to external subroutines. It affects neither the solution obtained nor the computed value of MUF.

¹NOSC TR 186, MINIMUF-3: A Simplified Hf MUF Prediction Algorithm, by RB Rose, JN Martin, and PH Levine, 1 February 1978

REVISED MINIMUF BASIC PROGRAM

A listing of the MINIMUF-3.5 program is included. Lines 100 through 720 contain a small driver which allows the model to be exercised. The actual MINIMUF program starts at line 1000.

The input variables for the MINIMUF program are as follows:

- L1 - Transmitter latitude, radians ($-\pi/2 \leq L1 \leq \pi/2$)
- W1 - Transmitter west longitude, radians ($-2\pi \leq W1 \leq 2\pi$)
- L2 - Receiver latitude, radians ($-\pi/2 \leq L2 \leq \pi/2$)
- W2 - Receiver west longitude, radians ($-2\pi \leq W2 \leq 2\pi$)
- M0 - Month ($1 \leq M0 \leq 12$)
- D6 - Day ($1 \leq D6 \leq 31$)
- T5 - Time (UT), hours ($0.0 \leq T5 \leq 24.0$)
- J9 - Output MUF, MHz
- S9 - Sunspot number
- P1 - 3.141593
- P0 - 1.570796

In using the test driver program, transmitter and receiver coordinates should be input in degrees; the driver converts these to radians as required by the MINIMUF subroutine.

Figure 1 is a sample output listing that users may find helpful in getting their version of MINIMUF to work.

```

TRANSMITTER LAT, LON = 21,156
RECEIVER LAT, LON = 38,122
DATE (DAY,MONTH) = 17,10
SUNSPOT NUMBER = 100

DATE: 17 OCT
TRANSMITTER LOCATION:  LATITUDE 21.00, LONGITUDE 156.00
RECEIVER LOCATION:    LATITUDE 38.00, LONGITUDE 122.00
SUNSPOT NUMBER = 110

      HOUR      MUF (MHZ)
      0         32.0
      1         32.0
      2         32.0
      3         29.9
      4         25.0
      5         22.0
      6         20.9
      7         19.3
      8         18.0
      9         16.9
     10         16.0
     11         15.2
     12         14.6
     13         14.1
     14         13.7
     15         21.0
     16         27.6
     17         31.5
     18         32.0
     19         32.0
     20         32.0
     21         32.0
     22         32.0
     23         32.0

PRESS RETURN TO PERFORM NEXT CASE.
```

Figure 1. Example of a 24-hour MUF listing of MINIMUF-3.5.

APPLICATION TIPS

With increasing solar activity, user interest in updating MINIMUF to reflect current solar activity has also increased. The updating method found to be most effective was to vary the sunspot number input parameter as a function of 107 mm solar radio flux. Because of the lag in F-region response to a rapid increase in solar activity, it is best to use either a 5-day, 15-day or 90-day running mean average of the 107 mm flux. The type of application will determine which is best. The 5-day mean is a shorter-term more dynamic input, while the 90-day mean is more applicable for long-term planning. These flux values can be acquired from

Space Environment Services Center, Boulder, Colorado

WWV transmissions at 18 minutes after each hour.

The conversion from 107 mm flux to sunspot number is accomplished by the graph shown as figure 2.

Two other points are borne out by field testing. First, MINIMUF is an F-region approximation. Any intervention by E-region modes of propagation, either as multiple E or EF complex modes, is not predictable on MINIMUF. Such operational situations, however, are proving to represent only a small percentage of the total. Second, MINIMUF has the greatest accuracy within the one- and two-hop ranges, between about 800 and 8000 km. Predictions for transmission paths longer than this should be used with some caution.




Figure 2. Relationship between smoothed mean Zurich sunspot number and smoothed mean 107 mm solar flux.

TEST DRIVER PROGRAM

```

100 INIT
110 DIM M$(37),A$(4),M(12)
120 DATA 31,28,31,30,31,30,31,31,30,31,30,31
130 READ M
140 M$="JANFEBMARAPR MAYJUNJUL AUGSEPOCTNOVDEC"
150 R0=PI/180
155 P1=2*PI
160 R1=180/PI
170 P0=PI/2
180 PAGE
190 PRINT "TRANSMITTER LAT, LON = ";
200 INPUT L1,W1
210 IF -90<=L1 AND L1<=90 THEN 240
220 PRINT "INVALID LATITUDE. MUST BE IN RANGE (-90,+90).";
230 GO TO 190
240 IF -360<=W1 AND W1<=360 THEN 270
250 PRINT "INVALID LONGITUDE. MUST BE IN RANGE (-360,+360).";
260 GO TO 190
270 PRINT "RECEIVER LAT, LON = ";
280 INPUT L2,W2
290 IF -90<=L2 AND L2<=90 THEN 320
300 PRINT "INVALID LATITUDE. MUST BE IN RANGE (-90,+90).";
310 GO TO 270
320 IF -360<=W2 AND W2<=360 THEN 350
330 PRINT "INVALID LONGITUDE. MUST BE IN RANGE (-360,+360).";
340 GO TO 270
350 PRINT "DATE (DAY,MONTH) = ";
360 INPUT D6,M0
370 IF 1<=M0 AND M0<=12 THEN 400
380 PRINT "INVALID MONTH. MUST BE IN RANGE (1,12).";
390 GO TO 350
400 IF 1<=D6 AND D6<=M(M0) THEN 430
410 PRINT USING 420:M(M0)
420 IMAGE "INVALID DAY. MUST BE IN RANGE (1, ",FD, ").";
425 GO TO 350
430 PRINT "SUNSPOT NUMBER = ";
440 INPUT S9
450 IF S9>0 THEN 480
460 PRINT "INVALID SUNSPOT NUMBER. MUST BE NON-NEGATIVE.";
470 GO TO 430
480 PAGE
490 A$=SEG(M$,3*M0-2,3)
500 PRINT USING ""DATE: ",FD,1X,FA":D6,A$
510 PRINT "TRANSMITTER LOCATION: ";
520 PRINT USING 530:L1,W1
530 IMAGE "LATITUDE ",FD.2D,", LONGITUDE ",FD.2D
540 PRINT "RECEIVER LOCATION: ";
550 PRINT USING 530:L2,W2
560 PRINT USING ""SUNSPOT NUMBER = ",FD":S9
570 PRINT
580 PRINT "    HOUR      MUF(MHZ)"
590 PRINT
600 L1=L1*R0
610 W1=W1*R0
620 L2=L2*R0
630 W2=W2*R0
640 FOR T5=0 TO 23
650 GOSUB 1000
660 PRINT USING 670:T5,J9
670 IMAGE 5X,2D,7X,2D.0
680 NEXT T5
690 PRINT
700 PRINT "PRESS RETURN TO PERFORM NEXT CASE.";
710 INPUT A$
720 GO TO 100

```

MINIMUF-3.5 PROGRAM

```

1000 REM
1010 K7=SIN(L1)*SIN(L2)+COS(L1)*COS(L2)*COS(W2-W1)
1020 G1=ACS(K7 MAX -1+1.0E-5 MIN 1-1.0E-8)
1030 K6=1.59*G1
1040 K6=K6 MAX 1
1050 K5=1/K6
1060 J9=100
1070 FOR K1=1/(2*K6) TO 1-1/(2*K6) STEP 0.9999-1/K6
1080 IF K5=1 THEN 1100
1090 K5=0.5
1100 P=SIN(L2)
1110 Q=COS(L2)
1120 A=(SIN(L1)-P*COS(G1))/(Q*SIN(G1))
1130 B=G1*K1
1140 C=P*COS(B)+Q*SIN(B)*A
1150 D=(COS(B)-C*P)/(Q*SQR(1-C*C))
1160 D=ACS(D MAX -1+1.0E-5 MIN 1-1.0E-8)
1170 W0=W2+SGN(SIN(W1-W2))*D
1180 IF W0=>0 THEN 1200
1190 W0=W0+P1
1200 IF W0<P1 THEN 1220
1210 W0=W0-P1
1220 L0=P0-ACS(C MAX -1+1.0E-5 MIN 1-1.0E-8)
1230 Y1=0.0172*(10+(W0-1)*30.4+D6)
1240 Y2=0.409*COS(Y1)
1250 K8=3.82*W0+12+0.13*(SIN(Y1)+1.2*SIN(2*Y1))
1260 K8=K8-12*(1+SGN(K8-24))*SGN(ABS(K8-24))
1270 IF COS(L0+Y2)>-0.26 THEN 1350
1280 K9=0
1290 G0=0
1300 M9=2.5*G1*K5
1310 M9=M9 MIN P0
1320 M9=SIN(M9)
1330 M9=1+2.5*M9*SQR(M9)
1340 GO TO 1590
1350 K9=(-0.26+SIN(Y2)*SIN(L0))/(COS(Y2)*COS(L0)+1.0E-3)
1360 K9=12-ATH(K9/SQR(ABS(1-K9*K9)))*7.639437
1370 T=K8-K9/2+12*(1-SGN(K8-K9/2))*SGN(ABS(K8-K9/2))
1380 T4=K8+K9/2-12*(1+SGN(K8+K9/2-24))*SGN(ABS(K8+K9/2-24))

```



```

1390 C0=ABS(COS(L0+Y2))
1400 T9=9.7*C0+9.6
1410 IF T9>0.1 THEN 1430
1420 T9=0.1
1430 M9=2.5*G1*K5
1440 M9=M9 MIN P0
1450 M9=SIN(M9)
1460 M9=1+2.5*M9*SQR(M9)
1470 IF T4<T THEN 1500
1480 IF (T5-T)*(T4-T5)>0 THEN 1510
1490 GO TO 1640
1500 IF (T5-T4)*(T-T5)>0 THEN 1640
1510 T6=T5+12*(1+SGN(T-T5))*SGN(ABS(T-T5))
1520 G9=PI*(T6-T)/K9
1530 G8=PI*T9/K9
1540 U=(T-T6)/T9
1550 G0=C0*(SIN(G9)+G8*(EXP(U)-COS(G9)))/(1+G8*G8)
1560 G7=C0*(G8*(EXP(-K9/T9)+1))*EXP((K9-24)/2)/(1+G8*G8)
1570 IF G0=>G7 THEN 1590
1580 G0=G7
1590 G2=(1+S9/250)*M9*SQR(6+58*SQR(G0))
1600 G2=G2*(1-0.1*EXP((K9-24)/3))
1610 G2=G2*(1+(1-SGN(L1)*SGN(L2))*0.1)
1620 G2=G2*(1-0.1*(1+SGN(ABS(SIN(L0))-COS(L0))))
1630 GO TO 1700
1640 T6=T5+12*(1+SGN(T4-T5))*SGN(ABS(T4-T5))
1650 G8=PI*T9/K9
1660 U=(T4-T6)/2
1670 U1=-K9/T9
1680 G0=C0*(G8*(EXP(U1)+1))*EXP(U)/(1+G8*G8)
1690 GO TO 1590
1700 IF G2>J9 THEN 1720
1710 J9=G2
1720 NEXT K1
1730 J9=J9 MAX 2 MIN 32
1740 RETURN

```


INITIAL DISTRIBUTION

HDQ, NAVAL SECURITY GROUP COMMAND
G53 (LT J FITTEN)

CHIEF OF NAVAL OPERATIONS
NOP-941F (HA FEIGLESON)

NAVAL COMMUNICATION UNIT,
WASHINGTON DC
NAVY ELECTROMAGNETIC SPECTRUM CENTER
CAPT JA MADIGAN

NAVAL AIR DEVELOPMENT CENTER
CODE 405 (WALTER SCHOPPE)

NATIONAL SECURITY AGENCY
R52 (WOODS)
R52 (T LEMKE)

USA CEEIA
CCC-EMEO-ED (GEORGE LANE)
FORT HUACHUCA, AZ 85613

MEGATEK CORP
1055 SHAFTER ST
SAN DIEGO, CA 92106
DR PH LEVINE

MITRE CORP
WESTGATE RESEARCH PARK
MC LEAN, VA 22101
DR ALLEN SCHNEIDER

GENERAL ELECTRIC COMPANY
BLDG 4, RM 48, COURT ST PLANT
SYRACUSE, NY 13201
DR GEORGE H MILLMAN

TRACOR, INC
1600 WILSON BLVD
ARLINGTON, VA 22209
DR THEODORE J COHEN

SOUTHWEST RESEARCH INSTITUTE
8500 CULEBRA DRIVE
PO DRAWER 28510
SAN ANTONIO, TX 78284
WM SHERRILL

COMMUNICATIONS RESEARCH CENTER
SHIRLEY BAY, BOX 490, STATION A
OTTAWA, ONTARIO, K1N8TS, CANADA
UI CAMPBELL

DEFENSE DOCUMENTATION CENTER (12)

END

FILMED

1-86

DTIC